<cfcomponent extends="org.mangoblog.plugins.BasePlugin">

	<cfset variables.package = "com/thecfguy/mango/plugins/mobiletheme"/>
	<!--- :::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::: --->

	<cffunction name="init" access="public" output="false" returntype="any">
		<cfargument name="mainManager" type="any" required="true" />
		<cfargument name="preferences" type="any" required="true" />
		<cfset super.init(arguments.mainManager, arguments.preferences) />
		<!--- OriginalTheme will store current setup theme while switching to mobile theme. --->
		<cfset initSettings(mobileThemeName='mobile',originalTheme='') />
		<cfreturn this/>
	</cffunction>

	<!--- :::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::: --->

	<cffunction name="setup" hint="This is run when a plugin is activated" access="public" output="false" returntype="any">
		<cfset super.setup() />
		<cfreturn "Plugin activated. <br /><a href='generic_settings.cfm?event=showMobileThemeSettings&amp;owner=thecfguy&amp;selected=showMobileThemeSettings'>Click Here</a> to select your mobile device theme?" />
	</cffunction>

	<!--- :::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::: --->

	<cffunction name="processEvent" hint="Synchronous event handling" access="public" output="false" returntype="any">
		<cfargument name="event" type="any" required="true" />
		<cfset var local = structnew() />
		<cfset local.mobileThemeName = getSetting('mobileThemeName') />
		
		<cfif listFindnocase("beforeIndexTemplate,beforeAuthorTemplate,beforeArchivesTemplate,beforePageTemplate,beforePostTemplate,beforeGenericTemplate,beforeOuputTemplate",arguments.event.name)>
			<cfif not len(local.mobileThemeName)><cfreturn arguments.event></cfif>
			<cfif structKeyExists(event.data.externalData,"isMobile")>
				<cfset session.isMobile = event.data.externalData.isMobile>
			<cfelseif reFindNoCase("android|avantgo|blackberry|blazer|compal|elaine|fennec|hiptop|iemobile|ip(hone|od)|iris|kindle|lge |maemo|midp|mmp|opera m(ob|in)i|palm( os)?|phone|p(ixi|re)\/|plucker|pocket|psp|symbian|treo|up\.(browser|link)|vodafone|wap|windows (ce|phone)|xda|xiino",CGI.HTTP_USER_AGENT) GT 0 OR reFindNoCase("1207|6310|6590|3gso|4thp|50[1-6]i|770s|802s|a wa|abac|ac(er|oo|s\-)|ai(ko|rn)|al(av|ca|co)|amoi|an(ex|ny|yw)|aptu|ar(ch|go)|as(te|us)|attw|au(di|\-m|r |s )|avan|be(ck|ll|nq)|bi(lb|rd)|bl(ac|az)|br(e|v)w|bumb|bw\-(n|u)|c55\/|capi|ccwa|cdm\-|cell|chtm|cldc|cmd\-|co(mp|nd)|craw|da(it|ll|ng)|dbte|dc\-s|devi|dica|dmob|do(c|p)o|ds(12|\-d)|el(49|ai)|em(l2|ul)|er(ic|k0)|esl8|ez([4-7]0|os|wa|ze)|fetc|fly(\-|_)|g1 u|g560|gene|gf\-5|g\-mo|go(\.w|od)|gr(ad|un)|haie|hcit|hd\-(m|p|t)|hei\-|hi(pt|ta)|hp( i|ip)|hs\-c|ht(c(\-| |_|a|g|p|s|t)|tp)|hu(aw|tc)|i\-(20|go|ma)|i230|iac( |\-|\/)|ibro|idea|ig01|ikom|im1k|inno|ipaq|iris|ja(t|v)a|jbro|jemu|jigs|kddi|keji|kgt( |\/)|klon|kpt |kwc\-|kyo(c|k)|le(no|xi)|lg( g|\/(k|l|u)|50|54|e\-|e\/|\-[a-w])|libw|lynx|m1\-w|m3ga|m50\/|ma(te|ui|xo)|mc(01|21|ca)|m\-cr|me(di|rc|ri)|mi(o8|oa|ts)|mmef|mo(01|02|bi|de|do|t(\-| |o|v)|zz)|mt(50|p1|v )|mwbp|mywa|n10[0-2]|n20[2-3]|n30(0|2)|n50(0|2|5)|n7(0(0|1)|10)|ne((c|m)\-|on|tf|wf|wg|wt)|nok(6|i)|nzph|o2im|op(ti|wv)|oran|owg1|p800|pan(a|d|t)|pdxg|pg(13|\-([1-8]|c))|phil|pire|pl(ay|uc)|pn\-2|po(ck|rt|se)|prox|psio|pt\-g|qa\-a|qc(07|12|21|32|60|\-[2-7]|i\-)|qtek|r380|r600|raks|rim9|ro(ve|zo)|s55\/|sa(ge|ma|mm|ms|ny|va)|sc(01|h\-|oo|p\-)|sdk\/|se(c(\-|0|1)|47|mc|nd|ri)|sgh\-|shar|sie(\-|m)|sk\-0|sl(45|id)|sm(al|ar|b3|it|t5)|so(ft|ny)|sp(01|h\-|v\-|v )|sy(01|mb)|t2(18|50)|t6(00|10|18)|ta(gt|lk)|tcl\-|tdg\-|tel(i|m)|tim\-|t\-mo|to(pl|sh)|ts(70|m\-|m3|m5)|tx\-9|up(\.b|g1|si)|utst|v400|v750|veri|vi(rg|te)|vk(40|5[0-3]|\-v)|vm40|voda|vulc|vx(52|53|60|61|70|80|81|83|85|98)|w3c(\-| )|webc|whit|wi(g |nc|nw)|wmlb|wonu|x700|xda(\-|2|g)|yas\-|your|zeto|zte\-",Left(CGI.HTTP_USER_AGENT,4)) GT 0>
				<cfset session.isMobile = 1>
			</cfif>
			<cfif session.isMobile>
				<cfset request.skin = local.mobileThemeName>
			</cfif>
		<cfelseif  arguments.event.name is "settingsNav">
			<cfset local.link = structnew() />
			<cfset local.link.owner = "thecfguy">
			<cfset local.link.page = "settings" />
			<cfset local.link.title = "Mobile Theme Switcher" />
			<cfset local.link.eventName = "showMobileThemeSettings" />
			<cfset arguments.event.addLink(local.link)>
		<cfelseif  arguments.event.name EQ "showMobileThemeSettings">
			<cfset local.data = arguments.event.getData() />
			<cfif structkeyexists(local.data.externaldata,"apply")>
				<cfset local.path = getManager().getBlog().getBasePath() />
				<cfif structkeyexists(local.data.externalData,"mobileThemeName")>
					<cfset setSettings(	mobileThemeName = local.data.externaldata.mobileThemeName) />
				<cfelse>
					<cfset setSettings(	mobileThemeName ='mobile') />
				</cfif>
				<cfset local.data.message.setstatus("success") />
				<cfset local.data.message.setType("settings") />
				<cfset local.data.message.settext("settings updated successfully") />
			</cfif>
			<cfsavecontent variable="local.page">
			<cfinclude template="admin/settingsForm.cfm">
		</cfsavecontent>
			<!--- change message --->
			<cfset local.data.message.setTitle("Mobile Theme") />
			<cfset local.data.message.setData(local.page) />
		</cfif>
		<cfreturn arguments.event />
	</cffunction>
</cfcomponent>
